
This case study looks at the re-engineering of a Fortran program to play rock-
paper-scissors.

SYNOPSIS
This case study looks at the re-engineering of a Fortran program to play rock-

paper-scissors.

Type: legacy re-engineering

Language: Fortran

Compiler: gfortran

Skills: Fortran re-engineering, re-writing legacy programs, 
language skills, removing goto’s

Experience Level: intermediate

Raj Koothrappali:  I'll tell you what. How about we go rock-paper-scissors? 

Sheldon Cooper: Ooh, I don't think so. Anecdotal evidence suggests that in the game of rock-paper-

scissors, players familiar with each other will tie 75 to 80% of the time due to the limited number of 

outcomes. I suggest rock-paper-scissors-lizard-Spock. 

Raj Koothrappali: What? 

Sheldon Cooper: It's very simple. Scissors cuts paper. Paper covers rock. Rock crushes lizard. Lizard 

poisons Spock. Spock smashes scissors. Scissors decapitates lizard. Lizard eats paper. Paper disproves 

Spock. Spock vaporizes rock. And as it always has, rock crushes scissors.

INTRODUCTION

This case study looks at converting a game of  Rock, Paper, Scissors (RPS) from an early dialect of 
Fortran (F77) to Fortran 90/95, removing legacy features1. RPS is a hand game which involves 
making a motion three times with your fists, and then either show a flat hand (paper), clenched 
fist (rock), or two fingers extended and separated (scissors). Depending on the results, the game is 
a tie (both show the same), or one person wins. Rock breaks scissors,  scissors cuts paper, paper 
covers rock. 

1 Adapted from a BASIC program  written by Charles Lund in BASIC Computer Games, Ahl, D.H. (ed). 

ORIGINAL FORTRAN PROGRAM

The program has a number of  legacy issues: (i) readability is hampered due to the fixed 
formatting and uppercase nature of  the code; (ii) label based loops, (iii) GO TO statements and 
(iv) arithmetic IFs.

      PROGRAM ROCKPS 
C
C     PROGRAM TO PLAY ROCK, PAPER, SCISSORS
C
      INTEGER X, K, N
      WRITE (*,*) "THIS PROGRAM ALLOWS YOU TO PLAY"
      WRITE (*,*) "ROCK, PAPER, SCISSORS"
  10  WRITE (*,*) "HOW MANY GAMES DO YOU WANT? " 
      READ (*,*) N
      IF (N .GT. 10) GO TO 10
      DO 160 I = 1, N
      WRITE (*,*) "GAME NUMBER", I
      X = CEILING(RAND()*3)
  15  WRITE (*,*) "3=ROCK,2=SCISSORS,1=PAPER"
      WRITE (*,*) "WHAT'S YOUR CHOICE? "
      READ (*,*) K
      IF ((K-1)*(K-2)*(K-3) .NE. 0) GO TO 15
      WRITE (*,*) "THIS IS MY CHOICE"
      GO TO (20,25,30), X         
  20  WRITE (*,*) "PAPER"; GO TO 50
  25  WRITE (*,*) "SCISSORS"; GO TO 50
  30  WRITE (*,*) "ROCK"
  50  IF (X.EQ.K) GO TO 155
      IF (X.GT.K) GO TO 125
      IF (X.EQ.1) GO TO 105
  95  WRITE (*,*) "YOU WIN!!"; H=H+1; GO TO 160
 105  IF (3-K) 95,115,95
 115  WRITE (*,*) "WOW, I WIN!"; C=C+1; GO TO 160
 125  IF (K.NE.1) GO TO 115 
 140  IF (3-X) 115,95,115
 155  WRITE (*,*) "TIE GAME - NO WINNER"
 160  CONTINUE 
 170  WRITE (*,*) "HERE IS THE FINAL SCORE"
 175  WRITE (*,*) "I HAVE WON ",C," GAME(S)"
 180  WRITE (*,*) "YOU HAVE WON ",H," GAME(S)"
 185  WRITE (*,*) "AND ",N-(C+H)," GAME(S) ENDED IN A TIE"
 190  WRITE (*,*) "THANKS FOR PLAYING!"
      END   

Here’s a sample program run.

 THIS PROGRAM ALLOWS YOU TO PLAY
 ROCK, PAPER, SCISSORS
 HOW MANY GAMES DO YOU WANT? 
2
 GAME NUMBER           1
 3=ROCK,2=SCISSORS,1=PAPER
 WHAT'S YOUR CHOICE? 
1
 THIS IS MY CHOICE
 PAPER
 TIE GAME - NO WINNER
 GAME NUMBER           2
 3=ROCK,2=SCISSORS,1=PAPER
 WHAT'S YOUR CHOICE? 
2
 THIS IS MY CHOICE
 PAPER
 YOU WIN!!
 HERE IS THE FINAL SCORE
 I HAVE WON    0.0000000      GAME(S)
 YOU HAVE WON    1.0000000      GAME(S)
 AND    1.0000000      GAME(S) ENDED IN A TIE
 THANKS FOR PLAYING!

WHAT ARE THE LEGACY ISSUES?

The first task is to assess the program, by disassembling it. Assume no prior knowledge, the task 
here is to re-engineer the program, not create it from scratch. Don’t be fooled by the programs 
mere 38 lines of  code. Re-engineering a program can be a challenging experience, no matter the 
size of  the program. Pull it apart and try and comprehend what is in the program.

The largest legacy issues with this program are the presence of  arithmetic if statements, and GO 
TO statements.  The arithmetic if  is a three-way conditional statement which appeared early on 
in Fortran II and was retained through Fortran IV, 66 and 77. It was the first conditional transfer 
statement available in Fortran II, before the advent of  the logical if  statement in Fortran IV., and 
the if-then-else in Fortran 77. The statement defines three different branches depending on 
whether the result of  the expression is negative, positive or zero. Consider the following example:

     integer b
     
     if (b) 20,10,30
  20 write(*,*) " if b < 0, control is transferred here"
  30 write(*,*) " if b > 0, control is transferred here"
  10 write(*,*) " if b = 0, control is transferred here"

This effectively says “if  the value of  b is less than 0, go to the statement at label 20”. Now 
consider the equivalent modern if  construct:

     if(b < 0) then
  !     write(*,*) "if b < 0, control is transferred here"
     else if (b > 0) then
         write(*,*) " if b > 0, control is transferred here"
     else
         write(*,*) " if b=0, control is transferred here"
     end if
      
One of  the problems with arithmetic if’s is that they facilitate a jump to occur, to a label 
somewhere in the program - unstructured coding. It essentially makes it extremely difficult to 
analyze a program. The other legacy issue deals with the presence of  go to statements - leading to 
further unstructured code. These go to statements effectively take the place of  loops. 

The legacy issues are highlighted in the code below.

1. A series of  six IF statements that branch using GO TO statements
2. A series of  four pure GO TO statements
3. A computed GO TO statement
4. A couple of  arithmetic IF statements

There is also a single DO loop which uses a label.

     PROGRAM ROCKPS 
C
C     PROGRAM TO PLAY ROCK, PAPER, SCISSORS
C
      INTEGER X, K, N
      WRITE (*,*) "THIS PROGRAM ALLOWS YOU TO PLAY"
      WRITE (*,*) "ROCK, PAPER, SCISSORS"
  10  WRITE (*,*) "HOW MANY GAMES DO YOU WANT? " 
      READ (*,*) N
      IF (N .GT. 10) GO TO 10
      DO 160 I = 1, N
      WRITE (*,*) "GAME NUMBER", I
      X = CEILING(RAND()*3)
  15  WRITE (*,*) "3=ROCK,2=SCISSORS,1=PAPER"
      WRITE (*,*) "WHAT'S YOUR CHOICE? "
      READ (*,*) K
      IF ((K-1)*(K-2)*(K-3) .NE. 0) GO TO 15
      WRITE (*,*) "THIS IS MY CHOICE"
      GO TO (20,25,30), X         
  20  WRITE (*,*) "PAPER"; GO TO 50
  25  WRITE (*,*) "SCISSORS"; GO TO 50
  30  WRITE (*,*) "ROCK"
  50  IF (X.EQ.K) GO TO 155
      IF (X.GT.K) GO TO 125
      IF (X.EQ.1) GO TO 105
  95  WRITE (*,*) "YOU WIN!!"; H=H+1; GO TO 160
 105  IF (3-K) 95,115,95
 115  WRITE (*,*) "WOW, I WIN!"; C=C+1; GO TO 160
 125  IF (K.NE.1) GO TO 115 
 140  IF (3-X) 115,95,115
 155  WRITE (*,*) "TIE GAME - NO WINNER"
 160  CONTINUE 
 170  WRITE (*,*) "HERE IS THE FINAL SCORE"
 175  WRITE (*,*) "I HAVE WON ",C," GAME(S)"
 180  WRITE (*,*) "YOU HAVE WON ",H," GAME(S)"
 185  WRITE (*,*) "AND ",N-(C+H)," GAME(S) ENDED IN A TIE"
 190  WRITE (*,*) "THANKS FOR PLAYING!"
      END

FIRST THINGS FIRST

The first thing to do is change the file extension to .f95, and modify the comment delimiter from 
C to !. The file will still compile and run.

See listing rockps.f95

IMPROVE READABILITY

To improve readability the first task is to convert the program from uppercase to lowercase. This 
can be achieved using the “translate characters” utility, tr:

! tr '[:upper:]' '[:lower:]' < input.txt > output.txt

Note that at the moment, the program is not format-free, this will be done after the jump labels 
have been removed, so there is some clarity during the re-engineering process. It can be indented 
as the re-engineering progresses to improve program clarity. Each major update will be provided 
in a separate file to illustrate the evolution of  the re-engineering process.

See listing rockps_1.f95

RE_ENGINEERING INPUT AND OUTPUT

This program has 38 lines of  code (35 without comments), of  which 20 lines of  code make up I/
O - 18 write statements and 2 read statements. There is nothing crazy about the I/O, so it will 
be left alone for now.

RE-ENGINEERING VARIABLES

The first portion of  the re-engineering involves modifying variables to remove implicit 
declarations, and enhance the declaration criteria to F95 standards. Variables that are implicitly 
declared in the original program are converted to explicit declarations. For example, arrays are 
converted int he following manner:

! integer x, k, n

becomes

! integer :: x, k, n

There are also two variables, c and h, that have no declaration, or values set to zero.

! integer :: x, k, n, c, h
! c = 0
! h = 0

RE-ENGINEERING OPERATORS

The old-fashioned operators such as .eq. should also be changed to their modern equivalent:

! .eq. becomes ==
! .gt. becomes >
! .ne. becomes /=
! .lt. becomes <

Note that .or., and .and. stay the same.

RE-ENGINEERING THE LEGACY ISSUES

So the question is which of  the legacy issues should be dealt with first?

1. A series of  six IF statements that branch using GO TO statements
2. A series of  four pure GO TO statements
3. A computed GO TO statement
4. A couple of  arithmetic IF statements
5. A single DO loop which uses a label.

The best place to start may be converting the DO loop, and work backwards from the issue with 
the least number of  incidents. 

RE-ENGINEERING THE DO LOOP

Modifying the DO loop is the easiest thing to do first. The old fashioned loop:

!  do 160 i = 1, n
!  ...
!  ...
!  ...
 160  continue 

easily becomes:

     do i = 1, n
 ! ...
! ...
! ...
     end do

If  however the code is compiled now, an error will occur because the label 160 has been removed 
and there are two co-dependent go to statements on lines with labels 95 and 115 reference 
label 160. The error is usually something of  the form:

! Error: Label 160 referenced at (1) is never defined

As both statements merely pass control to the next iteration of  the loop, it is easy to replace the 
go to statements with cycle statements (the equivalent of  continue in C). When the code hits 
a cycle statement, execution of  the current iteration of  the loop stops, and control returns to the 
top of  the loop. So this code:

!  do 160 i = 1, n
      ...
  95  write (*,*) "you win!!"; h=h+1; go to 160
      ...
 115  write (*,*) "wow, i win!"; c=c+1; go to 160
      ...
 160  continue 

becomes:

!  do 160 i = 1, n
      ...
  95  write (*,*) "you win!!"; h=h+1; cycle
      ...
 115  write (*,*) "wow, i win!"; c=c+1; cycle
      ...
 160  continue 

RE-ENGINEERING THE COMPUTED GO TO

Remember, re-engineering is not an exact science. Through analysis of  the program, decisions 
must be made as to which legacy issue is easiest to deal with. Removing these first usually makes 
the more complex things easier (or so the theory goes). The computed go to is associated with 
three labels, 20, 25, and 30.

In terms of  code:

      go to (20,25,30), x         
  20  write (*,*) "paper"; go to 50
  25  write (*,*) "scissors"; go to 50
  30  write (*,*) "rock"

This basically says that if  the value of  x is 1, 2, or 3, jump to label 20, 25, or 30 respectively. 
After the statements at each of  the labels have been processed, the program proceeds to the 
statement at label 50.  Without the go to statements at label 20 and 25, the programs logic 
would flow through to the next statement. For example, if  the code were to look like:

      go to (20,25,30), x         
  20  write (*,*) "paper"
  25  write (*,*) "scissors"
  30  write (*,*) "rock"

Then if  the value of  x were 1, the program would print “paper”, “scissors” and “rock”. The go 
to statements can be left intact for now. The computed go to can be re-engineered by turning 
it into an if-else sequence. So the new code looks like this:

      if (x == 1) then
          write (*,*) "paper"
      else if (x == 2) then
          write (*,*) "scissors"
      else if (x == 3) then
          write (*,*) "rock"
      end if

Notice something? The very nature of  the if-else sequence has made the remaining two go 
to statements redundant (which transfer control to label 50). The very act of  fixing the do loop 
and the computed go to have removed all four pure go to statements. Another side-effect of 
this is that the label 50 can also be removed, as it is now redundant.

See listing rockps_2.f95

RE-ENGINEERING THE IF STATEMENTS THAT USE GO TO TO BRANCH

Next in the firing line is are the six if statements that use go to statements to branch off. 

THE FIRST IF/GO TO

The first one in the program:

  10  write (*,*) "how many games do you want? " 
      read (*,*) n
      if (n .gt. 10) go to 10

This is effectively a loop of  sorts. If  the value of  n entered by the user is greater than 10, ask  the 
user to re-enter the value. If  it is suppose to be a loop, replace it with a loop. So the above code 
can be replaced with:

      do 
          write (*,*) "how many games do you want? " 
          read (*,*) n
          if (n .lt. 10) exit
      end do

Now the loop will continue asking for the number of  games until the user enters a value less than 
10, then the loop exits.

THE SECOND IF/GO TO

The next one in the program again has to do with input verification, and is effectively another 
quasi-loop:

  15      write (*,*) "3=rock,2=scissors,1=paper"
          write (*,*) "what's your choice? "
          read (*,*) k
          if ((k-1)*(k-2)*(k-3) .ne. 0) go to 15

It allows the user to enter a value for their choice, then does crazy math to decide if  the value 
entered is not a 1, 2, or 3. For example if  the user enters 4, then the if  statement calculates:

! (k-1)*(k-2)*(k-3) = (4-1)*(4-2)*(4-3)
!                   = 3 * 2 * 1
!                   = 6
!
So as 6 does not equal 0, control is transferred to label 15, and the process continues. A better 
way is just to add a loop of  the form:

! do 
!     write (*,*) "3=rock,2=scissors,1=paper"
!     write (*,*) "what's your choice? "
!     read (*,*) k
!     if (k==1 .or. k==2 .or. k==3) then
             exit
         end if
! end do

See listing rockps3.f95 

THE THIRD IF/GO TO

The third if-go to combination deals with the situation where both the user and computer 
choose the same object - effecting a tie:

          if (x.eq.k) go to 155
 ! ! ...
! ! ...
 155      write (*,*) "tie game - no winner"

This can easily be replaced with an actual if statement of  the form:

          if (x.eq.k) then
              write (*,*) "tie game - no winner"
              cycle
          end if

The statement at label 155 moves up into the if statement, which is now a more modern “if/
end if” sequence. The addition of  the cycle statement allows the program to continue to the 
next iteration of  the loop (i.e. next game, if  any).

See listing rockps4.f95

THE REMAINING IF/GO TO AND ARITHMETIC IF STATEMENTS

This is where things start to get trickier because the arithmetic if’s are embedded within the 
labels jumped to from the if statements. Here is the code, in all its spaghetti-ness:

          if (x.gt.k) go to 125
          if (x.eq.1) go to 105
  95      write (*,*) "you win!!"; h=h+1; cycle
 105      if (3-k) 95,115,95
 115      write (*,*) "wow, i win!"; c=c+1; cycle
 125      if (k.ne.1) go to 115 
 140      if (3-x) 115,95,115

This is best analyzed to see what is happening. The choices for the game are:

! rock = 3
! scissors = 2
! paper = 1

So obviously if  the player and computer choose the same value, it’s a tie, and has been dealt with. 
Winning involves the following rules for either computer or player.

! (i) 3>2 or (ii) 2>1 or (iii) 1>3

The first statement basically says that if  x (computer) is greater than k (player) jump to label 125 
(rule (i) or (ii)). At label 25, if  k!=1 (rule (i)), then the computer wins, otherwise it falls through to 
the next statement at label 140. At this point it calculates (3-x), and jumps according to the 
arithmetic if. If  (3-x) evaluates to -1,0, or 1 it jumps to labels 115, 95, or 115 respectively. 
So if  x=2, the program jumps to label 115 - the computer wins. 

Now the alternate scenario is if  x = 1, then the program jumps to label 105 (k could have either 
a value of  2 or 3) At this point it calculates (3-x), and jumps according to the arithmetic if. If 
(3-k) evaluates to -1,0, or 1 it jumps to labels 95, 115, or 95 respectively. So if  k=2, the 
program jumps to label 95, the player wins. If  k=3, the program jumps to label 115, the 
computer wins.

So far the program has dealt with:

! x=3, k=2 ! computer wins
! x=2, k=1 ! computer wins
! x=1, k=2 ! player wins
! x=1, k=3 ! computer wins

M.Wirth © 2014



The last two scenarios, which deal with the player winning occur when both if statements 
evaluate to false, and control passes to the statement at label 95, which signifies that the player 
wins.	

! x=3, k=1 ! player wins
! x=2, k=3 ! player wins

The problem with this code is that it is just convoluted and some scenarios are never realized. For 
example the statement at label 140 only deals with one scenario, when x=3 and k=2. Rather 
than trying to re-engineer this code, it makes much more sense to re-engineer the logic. Basically 
the code can become:

! if (((x==3).and.(k==2)).or. &
         ((x==2).and.(k==1)).or. &
         ((x==1).and.(k==3))) then
         write (*,*) "wow, i win!"
         c = c + 1
     else
         write (*,*) "you win!!"
         h = h + 1
     end if

Easier? I think so! That small piece of  re-engineered logic has taken care of  three if/go to 
combinations and both arithmetic ifs. All the labels have also been made redundant.

See listing rockps_5.f95

CLEANING UP READABILITY

So what is left now that the legacy features have been nixed! Firstly remove the remaining labels, 
which serve no purpose. There the following code:

 170  write (*,*) "here is the final score"
 175  write (*,*) "i have won ",c," game(s)"
 180  write (*,*) "you have won ",h," game(s)"
 185  write (*,*) "and ",n-(c+h)," game(s) ended in a tie"
 190  write (*,*) "thanks for playing!"

becomes:

      write (*,*) "here is the final score"
      write (*,*) "i have won ",c," game(s)"
      write (*,*) "you have won ",h," game(s)"
      write (*,*) "and ",n-(c+h)," game(s) ended in a tie"
      write (*,*) "thanks for playing!"

With that done, the program can be re-indented to fix any readability issues. 

See listing rockps_6.f95 

To improve readability even more, change the variable names to something more meaningful 
then single character identifiers.

! x ! compCH
! k ! playCH
! n ! nGames
! c ! compWin
! h ! playWin

See listing rockps_8.f95 

IMPROVING USABILITY

The last step involves the act of  improving the usability of  the game - not crucial, but playing it 
at the moment isn’t a very pleasant experience. This process really involves modifying the write 
statements to display the output better. 

See listing rockps_9.f95 

In certain cases, when a write is written like this:

! write (*,*) "How many games would you like to play? "
! read (*,*) nGames

the output is of  the form

! How many games would you like to play? 
! 2

It would be better to have the user input (2 in this case), at the end of  the prompt, as in:

! How many games would you like to play? 2

This is achieved by modifying the write statement:

! write (*,"(A)",advance="no") "How many games would you like to play? " 
! read (*,*) nGames

The inline formatting statement "(A)",advance="no" has the effect of  printing an ASCII 
string, and then not advancing, i.e. not outputting a <return>. In other write statements, inline 
formatting has been included to clean up the amount of  white spaces produced in the output. For 
example:

! write (*,*) "I have won ",compWin," game(s)"

becomes

! write (*,"(A,I1,A)") "I have won ",compWin," game(s)"

The formatting statement "(A,I1,A)" has the effect of  printing an ASCII string followed by 
an integer of  length 1, followed by an ASCII string. The / symbol in a formatting statement has 
the effect of  adding an empty newline (to separate sections, and make the output more readable). 
Here is the output of  the program now:

This program allows you to play: rock, paper, scissors
How many games would you like to play? 2
Game number: 1
Select from: rock=3, scissors=2, paper=1
what's your choice? 1
I chose paper
RESULT: Tie game - no winner

Game number: 2
Select from: rock=3, scissors=2, paper=1
what's your choice? 2
I chose paper
RESULT: You win!!

Final score
I have won 0 game(s)
you have won 1 game(s)
and 1 game(s) ended in a tie
Thanks for playing!

THE FINAL WORD

The program has increased in size from 35 to 48 LOC, implying a 40% increase in the number 
of  lines of  code. This is not unusual in the context of  re-engineering a program an removing 
legacy code.

